

4.1 设计 Oberon-0 语言的翻译模式

4.1.1 改造文法

4.1.1.1 消除二义性

- 包括递归下降在内的任何翻译模式都应该消除二义性
- 在ex1已经讨论过，可以认为Oberon-0是没有二义性的，所以这部分不需要任何改造

4.1.1.2 消除空产生式

- 将每个产生式右部的非终结符均替换成 ϵ ，如果该非终结符能推出 ϵ 的话
- 比如考虑下面ex3中 `procedure_call` 对应的产生式，`actual_parameters` 能推出 ϵ

```
1  procedure_call ::= IDENTIFIER: procedureName actual_parameters: actualParameters
2                  {:
3                  ...
4                  :}
5                  ;
6
7  actual_parameters ::= /* epsilon */
8                    | LeftParenthesis expression_list:parameters RightParenthesis
9                    {:                                     :}
10                   ;
```

- 把所有可能性都替换出来
- 比如考虑代入 `actual_parameters` 推出 ϵ 后的右部

```
1  procedure_call ::= IDENTIFIER: procedureName actual_parameters: actualParameters
2                  {:
3                  ...
4                  :}
5                  | IDENTIFIER: procedureName
6                  {:
7                  ...
8                  :}
9                  ;
10
11 actual_parameters ::= LeftParenthesis expression_list:parameters RightParenthesis
12                   {:                                     :}
13                   ;
```

- 但是本次实验不考虑消除空产生式：
 - 一方面这样修改文法的开销太大，需要递归地考虑所有可能推出空产生式的非终结符、再递归地将其代入其所在的所有位置，同时有可能会降低文法的可读性太差
 - 另一方面递归下降预测分析法可以处理空产生式、只需要额外定义一些其他的附加行为，但本质上产生式和语义动作的选择还是基于 `lookahead`，并不影响分析过程的正确性

4.1.1.3 消除直接左递归

- 对于具有直接左递归的产生式，其一般形式为：
 - $A \rightarrow A\alpha_1 \mid A\alpha_2 \mid \dots \mid A\alpha_m \mid \beta_1 \mid \beta_2 \mid \dots \mid \beta_n$
- 可以将这组产生式替换为以下两组等价的、无直接左递归的产生式：
 - $A \rightarrow \beta_1 A' \mid \beta_2 A' \mid \dots \mid \beta_n A'$
 - $A' \rightarrow \alpha_1 A' \mid \alpha_2 A' \mid \dots \mid \alpha_m A' \mid \epsilon$
- 比如本次实验中 `simple_expression` 和 `term` 对应的BNF表达式
 - `simple_expression = ["+"|" -"] term {"+"|" -"} term`
 - `term = factor {"*"|"DIV"} factor`
- 如果直接将上述BNF翻译为CFG的产生式，其中为了进行左结合而做出的设计就会导致左递归：

- 对ex3中LALR(1)分析法中不会引发问题、因为自底向上分析法对左递归不敏感
- 对递归下降预测分析法影响很大，会出现死循环

```

1  simple_expression ::= /*["+" | "-"] term {"+" | "-" | "OR") term} */
2                      term : termToken  term_list : termListToken
3                      {
4                          ...
5                      :}
6                      | ...
7                      ;
8
9  term ::= factor : factorToken { : ... :}
10         | term : termToken TIMES factor : factorToken
11         {
12             ...
13         :}
14         |

```

- 根据上述规则，将上面的BNF表达式改造为不具有左递归的产生式、转而成为右递归
 - `simple_expression -> [+|-] term post_term | term post_term`
 - `post_term -> (+|-|or) term post_term | ε`
 - `term -> factor post_factor`
 - `post_factor -> (*|&|div|mod) factor post_factor | ε`
- 其他直接左递归的处理方法也类似

4.1.1.4 消除间接左递归

- 对于具有间接左递归的产生式，其左递归需要将某个产生式代入另一个产生式的右部
- 经过下面算法代入后，可以将这一组产生式中所有间接左递归转化为直接左递归，实际上处理方法和消除直接左递归一致

Remove Left Recursion[消除左递归]



中山大學
SUN YAT-SEN UNIVERSITY

◆ **Step 1** : Arrange all non-terminals of grammar G in some order A_1, A_2, \dots, A_n ;

◆ **Step 2** : Execute in order obtained in Step 1:

FOR i:=1 TO n DO

FOR j:=1 TO i-1 DO

Replace each production of the form $A_i \rightarrow A_j \gamma$ by the productions

$A_i \rightarrow \delta_1 \gamma \mid \delta_2 \gamma \mid \dots \mid \delta_k \gamma$, where $A_j \rightarrow \delta_1 \mid \delta_2 \mid \dots \mid \delta_k$ are all current

A_j -productions;

[间接左递归问题 转换为 直接左递归问题]

eliminate the immediate left recursion among the A_i -productions;

[解决直接左递归问题]

END

END

$A \rightarrow \alpha \alpha' \mid \beta$

$A \rightarrow \beta \alpha'; \alpha' \rightarrow \alpha \alpha' \mid \epsilon$

◆ **Step 3** : Simplify the grammar obtained from Step 2 --- remove the production rules of non-terminal that can never be reached from the start symbol.

[去掉无效 productions]

20

4.1.1.5 提取左公因子

- 对于一组产生式，如果它们有共同的左公因子，其一般形式为：
 - $A \rightarrow \alpha \beta_1 \mid \alpha \beta_2 \mid \dots \mid \alpha \beta_n \mid \gamma$
- 可以将这组产生式替换为以下两组等价的、无公共左公因子的产生式：
 - $A \rightarrow \alpha A' \mid \gamma$
 - $A' \rightarrow \beta_1 \mid \beta_2 \mid \dots \mid \beta_n$
- 需要注意这里的操作可能是递归的，即如果替换后，新的 A' 产生式仍然有公共左因子，则需要重复上述过程
- 比如本次实验中 `statement`、`assignment` 和 `procedure_call` 对应的BNF表达式
 - `statement = [assignment | procedure_call | ...]`

- `assignment = identifier ...`
- `procedure_call = identifier ...`
- 如果直接将上述BNF翻译为CFG的产生式，会产生左公因子问题：
- 对ex3中LALR(1)分析法中不会引发问题、因为自底向上分析法对左公因子不敏感
- 对递归下降预测分析法影响较大，尽管可以通过回溯机制让递归下降预测分析法可以处理左公因子，但是这种操作开销过大，故需要考虑消除左公因子

```

1  statement ::= assignment
2              | procedure_call
3              | ...
4              ;
5
6  assignment ::= IDENTIFIER:identifier selector:selectorName ASSIGNMENT expression:expressionToken
7              { :
8              ...
9              : }
10             ;
11
12 procedure_call ::= IDENTIFIER: procedureName actual_parameters: actualParameters
13                { :
14                ...
15                : }
16                ;

```

- 根据上述规则，将上面的BNF表达式改造为不具有左公因子的产生式
 - `statement -> identifier id_statement | if_statement | while_statement`
 - `id_statement -> assignment | procedure_call`
- 其他左公因子的处理方法也类似

4.1.1.6 修改后的文法

```

    module → module identifier ; declarations statement_block end identifier .
statement_block → begin statement_sequence
                | ε
    declarations → const_declaration type_declaration var_declaration procedure_declaration_sequence
const_declaration → const const_assign
                | ε
    const_assign → identifier = expression ; const_assign
                | ε
    type_declaration → type type_assign
                | ε
    type_assign → identifier = type ; type_assign
                | ε
    type → identifier
        | type_arr
        | type_rec
        | integer
        | boolean
    type_arr → array expression of type
    type_rec → record field_list end
    field_list → identifier : type ; field_list
                | ε
    var_declaration → var var_list
                | ε
    var_list → identifier_list : type ; var_list
                | ε
    procedure_declaration_sequence → procedure_declaration ; procedure_declaration_sequence
                | ε
    procedure_declaration → procedure_head ; procedure_body
    procedure_head → procedure identifier procedure_parameters
    procedure_parameters → ( parameter_list )
                | ε
    parameter_list → parameter extended_parameter
    parameter → var identifier_list : type
                | identifier_list : type
    extended_parameter → ; parameter extended_parameter
                | ε
    identifier_list → identifier extended_identifier
    extended_identifier → , identifier extended_identifier
                | ε
    procedure_body → declarations statement_block end identifier
    statement_sequence → statement extended_statement
    extended_statement → ; statement extended_statement
                | ε
    statement → identifier identifier_statement
                | if_statement
                | while_statement
    identifier_statement → assignment
                | procedure_call
    assignment → selector := expression
    selector → . identifier selector
                | [ expression ] selector
                | ε
    procedure_call → act_parameters
    act_parameters → ( expression_list )
                | ε
    expression_list → expression extended_expression
                | ε
    extended_expression → , expression extended_expression
                | ε
    if_statement → if expression then statement_sequence elif_sequence else statement end
    elif_sequence → elsif expression then statement_sequence elif_sequence
                | ε
    else_statement → else statement_sequence
                | ε
    while_statement → while expression do statement_sequence end
    expression → simple_expression post_expression

```

$$\begin{aligned}
post_expression &\rightarrow rel_op\ simple_expression \\
&| \epsilon \\
rel_op &\rightarrow = | \# | < \\
&| <= | > | >= \\
simple_expression &\rightarrow +\ term\ post_term \\
&| -\ term\ post_term \\
&| \ term\ post_term \\
post_term &\rightarrow +\ term\ post_term \\
&| -\ term\ post_term \\
&| \text{or}\ term\ post_term \\
&| \epsilon \\
term &\rightarrow factor\ post_factor \\
post_factor &\rightarrow * factor\ post_factor \\
&| \&\ factor\ post_factor \\
&| \text{div}\ factor\ post_factor \\
&| \text{mod}\ factor\ post_factor \\
&| \epsilon \\
factor &\rightarrow \text{identifier}\ selector \\
&| \text{num_token} \\
&| (expression) \\
&| \sim factor
\end{aligned}$$

4.1.2 翻译模式

4.1.2.1 数据结构

4.1.2.1.1 env 符号表

- 符号表类储存当前作用域的各种标识符
- 符号表需要使用 `new Env(env)` 的语义动作支持嵌套作用域的栈式结构
- 当进入一个新的过程作用域，应该创建一个新的 `env` 并将其链接到旧的 `env`
- 查找符号时，会先在当前作用域对应的 `env` 查找，如果找不到，则沿着链接去父作用域对应的 `env` 查找

4.1.2.1.2 Decl 声明

- 声明类存储标识符的名称、类型 `type` 和 `const`、`var` 等修饰符

4.1.2.1.3 Type 类型

- 类型层次结构，包括 `IntegerType`、`BooleanType`、`ArrayType`、`RecType`、`FunctionType`、`ModuleType` 等
- 类型结构主要用于类型检查等语义分析过程

4.1.2.1.4 g_mod, g_proc

- 调用图的数据结构，对应调用图
- `g_mod` 代表整个模块
- `g_proc` 代表其中的一个过程。

4.1.2.1.5 calls

- 全局调用列表，可以汇总解析时遇到的所有过程调用
- 显然在调用一个过程时，可能出现前向引用，此时过程的完整定义可能还未被解析，所以不能立即验证分析
- 全局调用列表将验证工作推迟到整个过程解析完毕后

4.1.2.1.6 各种属性

- 非终结符具有综合属性和继承属性，终结符也有综合属性、但是由词法分析器给定不需要额外分析
 - 综合属性是自底向上传递信息，一般由其子节点的属性
 - 继承属性是自顶向上传递信息，一般自上向下传递
-

4.1.2.2 语义动作

4.1.2.2.1 作用域分析

- `module` 规则
 - 初始化 `g_mod` 作为顶层流图，初始化全局符号表
 - 使用 `{ cur = env; env = new Env(env); }` 表示进入模块作用域
 - 使用 `{ g_proc = g_mod.add('Main'); stat_block.g_stats = g_proc::add }` 表示：
 - `g_mod` 中创建过程流图 `g_proc`、命名为 `main`，用于存放主程序体的语句
 - `g_proc` 的 `add` 方法作为继承属性 `g_stats` 传递给 `stat_block`
 - 上述操作将主程序中所有过程都添加到主模块中
 - `{ env = cur }` 表示退出模块作用域
- `proc` 规则
 - 初始化 `g_proc` 作为局部流图，初始化局部符号表
 - `{ proc_body.g_proc = g_mod.add(proc_head.id.lexeme) }` 表示在全局模块流图 `g_mod` 中，为当前过程创建一个新的过程流图 `g_proc`
 - `{ stat_block.g_stats = proc_body.g_proc::add }` 表示将新创建的 `g_proc` 的 `add` 方法传递下去，确保该过程体内的语句被正确地添加到自己的流图，其中 `add` 是 java 提供的接口 `interface`

4.1.2.2.2 流图构建

- `stat_seq` 规则
 - `{ stat.g_stats = stat_seq.g_stats }` 表示将从父节点接收到的 `g_stats` 抽象函数接口直接传递给它的子节点 `stat`
 - 在流图上形成了一条上下文传递链
- `stat` 规则
 - `{ stat.g_stats.add(if_stat.statement) }` 表示 `if_stat` 完成解析后，编译器会向上返回完整的 `IfStatement` 流图节点对象实例
 - `stat` 节点接收到这个节点后，调用继承下来的 `g_stats` 函数，将这个 `IfStatement` 节点添加到当前上下文的语句列表中
- `if_stat` 规则
 - `{ statement = new IfStatement(expr.text) }`: 遇到 `if`，立即创建一个 `IfStatement` 流图节点。这个节点内部有自己的 `"true"` 和 `"false"` 语句列表。
 - `{ g_tbody = statement.getTrueBody(); stat_seq.g_stats = g_tbody::add }` 表示
 - 首先获取 `IfStatement` 节点的 `true` 分支列表 `g_tbody`
 - 然后切换上下文，将 `g_tbody` 的 `add` 方法作为新的 `g_stats` 继承属性，传递给 `then` 后面的 `stat_seq`
 - 最后 `then` 代码块中的所有语句都会被添加到 `IfStatement` 节点的 `true` 分支中，而不是外部语句列表当中
 - `{ g_fbody = statement.getFalseBody(); elif_seq.g_stats = g_fbody::add }` 表示
 - 首先获取 `IfStatement` 节点的 `false` 分支列表 `g_fbody`
 - 然后切换上下文，将 `g_fbody` 的 `add` 方法作为新的 `g_stats` 继承属性，传递给 `elif`、`else` 后面的 `stat_seq`
 - 最后 `then` 代码块中的所有语句都会被添加到 `IfStatement` 节点的 `false` 分支中，而不是外部语句列表当中
 - `{ if_stat.statement = statement }` 表示将构建完成的、包含了所有内嵌语句的 `IfStatement` 节点作为综合属性向上返回
- `while_stat` 规则
 - 在本质上也是通过 `true` 分支和 `false` 分支进行语义分析
 - 工作原理与上述 `if_stat` 完全相同，不再过多赘述

4.1.2.2.3 表达式、赋值、过程调用

- `expr`、`term`、`factor` 表达式规则
 - 语义动作是类型检查和文本合成
 - 语义检查确保表达式中的类型兼容，并生成一个文本表示
 - 文本表示可以被放入 `PrimitiveStatement`、`IfStatement` 等流图节点中作为对应的标签
- `assignment`、`proc_call` 规则
 - `{ assignment.statement = new PrimitiveStatement(...) }` 表示创建 `PrimitiveStatement` 对象
 - `PrimitiveStatement` 对象是一个流图节点，内容是赋值或过程调用的文本表示

- `PrimitiveStatement` 对象随后通过 `id_stat` 规则, 被 `id_stat.g_stats.add(...)` 语义动作添加到当前的流图分支中

4.1.2.3 最终翻译模式

```

s → { env = null }
module
{ for ( call in calls ) call.verify() }
{ module.g_mod.show() }
module → module id ;
{ decl = new Decl(id, new ModuleType()); env.put(decl) }
{ g_mod = new Module(id.lexeme); module.g_mod = g_mod }
{ cur = env; env = new Env(env); }
{ decls.g_mod = g_mod; }
decls
{ g_proc = g_mod.add( 'Main' ); stat_block.g_stats = g_proc :: add }
stat_block end id .
{ env = cur }
stat_block → begin
{ stat_seq.g_stats = stat_block.g_stats }
stat_seq
| ε
decls → cnst_decl type_decl var_decl
{ proc_decl_seq.g_mod = decls.g_mod }
proc_decl_seq
cnst_decl → const cnst_ass
| ε
cnst_ass → id = expr
{ decl = new Decl(id, expr.type, { modifiers : [ Decl.Modifiers.Const ] }) }
{ env.put(id.lexeme, decl) }
; cnst_ass1
| ε
type_decl → type type_ass
| ε
type_ass → id = type
{ decl = new Decl(id, new Typedef(type.val)) }
{ env.put(id.lexeme, decl) }
; type_ass1
| ε
type → id
{ decl = env.get(id.lexeme); type.val = decl.type.val }
| type_arr
{ type.val = type_arr.val }
| type_rec
{ type.val = type_rec.val }
| integer
{ type.val = new IntegerType() }
| boolean
{ type.val = new BooleanType() }
type_arr → array expr of type
{ type_arr.val = new ArrayType(type.val) }
type_rec → record
{ field_list.l_fields = { } }
field_list end
{ type_rec.val = new RecType(field_list.fields) }
field_list → id : type ;
{ decl = new Decl(id, type.val, { modifiers : [ Decl.Modifiers.Var ] }) }
{ field_list.l_fields.put(id.lexeme, decl); field_list1.l_fields = field_list.l_fields }
field_list1
{ field_list.fields = field_list1.fields }
| ε { field_list.fields = field_list.l_fields }
var_decl → var var_list
| ε
var_list → id_list : type ;
{
for(id in id_list.ids) {
env.put(id.lexeme, new Decl(id, type.val, { modifiers : [ Decl.Modifiers.Var ] }));
}
}
var_list1
| ε

```

```

proc_decl_seq → { proc_decl.g_mod = proc_decl_seq.g_mod }
proc_decl ;
{ proc_decl_seq1.g_mod = proc_decl_seq.g_mod }
proc_decl_seq1
| ε
proc_decl → { cur = env; env = new Env(env) }
proc_head ;
{ proc_body.g_mod = proc_decl.g_mod; proc_body.g_proc = proc_decl.g_mod.add(proc_head.id.lexeme) }
{ proc_body.decl_id = proc_head.id }
proc_body
{ env = cur; env.put(proc_head.id, proc_head.decl) }
proc_head → procedure id proc_pars
{ decl = new Decl(id, new FunctionType(proc_pars.params)) }
{ proc_head.decl = decl; proc_head.id = id }
proc_pars → ( par_list ) { proc_pars.params = par_list.params }
| ε { proc_pars.params = [] }
par_list → par ext_par
{ par_list.params = [ ...par.params, ...ext_par.params ] }
par → { par.params = [] }
var id_list : type
{
  for(id in id_list.ids) {
    decl = new Decl(id, type.val, { modifiers : [ Decl.Modifiers.Var ] });
    par.params.push(decl);
    env.put(id.lexeme, decl);
  }
}
| { par.params = [] }
id_list : type
{
  for(id in id_list.ids) {
    decl = new Decl(id, type.val);
    par.params.push(decl);
    env.put(id.lexeme, decl);
  }
}
ext_par → ; par ext_par1
{ ext_par.params = [ ...par.params, ...ext_par1.params ] }
| ε { ext_par.params = [] }
id_list → id ext_id { id_list.ids = [ id, ...ext_id.ids ] }
ext_id → , id ext_id1 { ext_id.ids = [ id, ...ext_id1.ids ] }
| ε { ext_id.ids = [] }
proc_body → { decls.g_mod = proc_body.g_mod }
decls
{ stat_block.g_stats = proc_body.g_proc :: add }
stat_block end id
stat_seq → { stat.g_stats = stat_seq.g_stats }
stat
{ ext_stat.g_stats = stat_seq.g_stats }
ext_stat
ext_stat → ; { stat.g_stats = ext_stat.g_stats }
stat
{ ext_stat1.g_stats = ext_stat.g_stats }
ext_stat1
| ε
stat → id
{ id_stat.g_stats = stat.g_stats; id_stat.id = id }
id_stat
| if_stat { stat.g_stats.add(if_stat.statement) }
| while_stat { stat.g_stats.add(while_stat.statement) }
id_stat → { assignment.decl = env.get(id_stat.id.getLexeme()); assignment.id = id_stat.id }
assignment
{ id_stat.g_stats.add(assignment.statement) }
| { proc_call.decl = env.get(id_stat.id.getLexeme()); proc_call.id = id_stat.id }
proc_call

```

```

      { id_stat.g_stats.add(proc_call.statement) }
assignment → { decl = assignment.decl; sel.parent_type = decl.type }
              sel := expr
              { assignment.statement = new PrimitiveStatement(decl.lexeme || sel.text || ' := ' || expr.text) }
sel → .id
      { type = sel.parent_type.fields.get(id); sel1.parent_type = type }
      sel1
      { sel.text = '.' || id.lexeme || sel1.text; sel.type = sel1.type }
      | [ expr ]
      { sel1.parent_type = sel.parent_type.type }
      sel1
      { sel.text = '[' || expr.text || ']' || sel1.text; sel.type = sel1.type }
      | ε { sel.text = ""; sel.type = sel.parent_type }
proc_call → { decl = proc_call.decl }
            act_pars
            { calls.add(new ProcCall(proc_call.id, act_pars.types)) }
            { proc_call.statement = new PrimitiveStatement(decl.lexeme || act_pars.text) }
act_pars → ( expr_list )
           { act_pars.types = expr_list.types }
           { act_pars.text = '(' || expr_list.text || ')' }
           | ε
           { act_pars.types = [] }
           { act_pars.text = "" }
expr_list → expr ext_expr
           { expr_list.types = [ expr.type, ...ext_expr.types ] }
           { expr_list.text = expr.text || ext_expr.text }
           | ε
           { expr_list.types = [] }
           { expr_list.text = "" }
ext_expr → , expr ext_expr
          { ext_expr1.types = [ expr.type, ...ext_expr.types ] }
          { ext_expr.text = ',' || expr.text || ext_expr1.text }
          | ε
          { ext_expr.types = [] }
          { ext_expr.text = "" }
if_stat → if expr
         { statement = new IfStatement(expr.text) }
         then
         { g_tbody = statement.getTrueBody(); stat_seq.g_stats = g_tbody :: add }
         stat_seq
         { g_fbody = statement.getFalseBody(); elif_seq.g_stats = g_fbody :: add }
         elif_seq
         { g_stats = elif_seq.g_pstats; else_stat.g_stats = g_stats }
         else_stat end
         { if_stat.statement = statement }
elif_seq → elsif expr
          { statement = new IfStatement(expr.text) }
          then
          { g_tbody = statement.getTrueBody(); stat_seq.g_stats = g_tbody :: add }
          stat_seq
          { g_fbody = statement.getFalseBody(); elif_seq1.g_stats = g_fbody :: add }
          elif_seq1
          { elif_seq.g_stats.add(statement); elif_seq.g_pstats = elif_seq1.g_pstats }
          | ε { elif_seq.g_pstats = elif_seq.g_stats }
else_stat → else
           { stat_seq.g_stats = else_stat.g_stats }
           stat_seq
           | ε
while_stat → while expr
            { statement = new WhileStatement(expr.text) }
            do
            { g_lbody = statement.getLoopBody(); stat_seq.g_stats = g_lbody :: add }
            stat_seq end
            { while_stat.statement = statement }
expr → simple_expr
      { next_expr.type = simple_expr.type }

```

```

    { post_expr.l_type = simple_expr.type }
    { post_expr.l_text = simple_expr.text }
    post_expr
    { expr.type = post_expr.type }
    { expr.text = post_expr.text }
post_expr → rel_op simple_expr
    { post_expr.text = post_expr.l_text || rel_op.text || simple_expr.text }
    { post_expr.type = new BooleanType() }
| ε
    { post_expr.text = post_expr.l_text }
    { post_expr.type = post_expr.l_type }
rel_op → = { rel_op.text = '=' }
| # { rel_op.text = '#' }
| < { rel_op.text = '<' }
| <= { rel_op.text = '<=' }
| > { rel_op.text = '>' }
| >= { rel_op.text = '>=' }
simple_expr → + term
    { post_term.l_type = new IntegerType() }
    { post_term.l_text = '(' || term.text || ')' }
    post_term
    { simple_expr.type = post_term.type; simple_expr.text = post_term.text }
| - term
    { post_term.l_type = new IntegerType() }
    { post_term.l_text = '-' || term.text || ')' }
    post_term
    { simple_expr.type = post_term.type; simple_expr.text = post_term.text }
| term
    { post_term.l_type = term.type; post_term.l_text = term.text }
    post_term
    { simple_expr.type = post_term.type; simple_expr.text = post_term.text }
post_term → + term
    { post_term1.l_type = new IntegerType() }
    { post_term1.l_text = post_term.l_text || '+' || term.text }
    post_term1
    { post_term.type = post_term1.type; post_term.text = post_term1.text }
| - term
    { post_term1.l_type = new IntegerType() }
    { post_term1.l_text = post_term.l_text || '-' || term.text }
    post_term1
    { post_term.type = post_term1.type; post_term.text = post_term1.text }
| or term
    { post_term1.l_type = new BooleanType() }
    { post_term1.l_text = post_term.l_text || 'or' || term.text }
    post_term1
    { post_term.type = post_term1.type; post_term.text = post_term1.text }
| ε
    { post_term.type = post_term.l_type }
    { post_term.l_text = post_term.l_text }
term → factor
    { post_factor.l_type = factor.type; post_factor.l_text = factor.text }
    post_factor
    { term.type = post_factor.type; term.text = post_factor.text }
post_factor → * factor
    { post_factor1.l_type = new IntegerType() }
    { post_factor1.l_text = post_factor.l_text || '*' || factor.text }
    post_factor1
    { post_factor.type = post_factor1.type; post_factor.text = post_factor1.text }
| & factor
    { post_factor1.l_type = new BooleanType() }
    { post_factor1.l_text = post_factor.l_text || '&' || factor.text }
    post_factor1
    { post_factor.type = post_factor1.type; post_factor.text = post_factor1.text }
| div factor
    { post_factor1.l_type = new IntegerType() }
    { post_factor1.l_text = post_factor.l_text || '/' || factor.text }

```

```

    { post_factor1.l_text = post_factor.l_text || 'div' || factor.text }
    post_factor1
    { post_factor.type = post_factor1.type; post_factor.text = post_factor1.text }
  | mod factor
    { post_factor1.l_type = new IntegerType() }
    { post_factor1.l_text = post_factor.l_text || 'mod' || factor.text }
    post_factor1
    { post_factor.type = post_factor1.type; post_factor.text = post_factor1.text }
  | ε
    { post_factor.type = post_factor.l_type }
    { post_factor.text = post_factor.l_text }
factor → id
    { decl = env.get(id.lexeme); sel.parent_type = decl.type }
    sel
    { factor.type = sel.type; factor.text = id.lexeme || sel.text }
| num_token
    { factor.type = new IntegerType(); factor.text = num_token.lexeme }
| ( expr )
    { factor.type = expr.type; factor.text = '(' || expr.text || ')' }
| ~ factor1
    { factor.type = new BooleanType(); factor.text = '~' || factor1.text }

```

4.2 编写递归下降预测分析程序

4.2.1 词法分析

- 词法分析部分与ex2一致
- 对上面已经给出 Oberon-0 语言词法规则的正则定义式，可以根据上面给出的正则式给出jflex的代码

```

1  /* Regular definitions */
2
3  MyInteger      = 0[0-7]* | [1-9]+[0-9]*
4  Identifier     = [a-zA-Z][a-zA-Z0-9]*
5  whitespace    = [ \t\n\r] | \r\n
6  Comment       = "(*" ~ "*)"

```

- 对其他不需要正则表达式的单词，比如 MODULE、>=，可以直接使用字符串匹配，jflex也可以很简单地处理字符串匹配

```

1  "MODULE"      { return Keyword.symModule(yyline + 1, yycolumn + 1); }
2
3  ">="          { return Operator.gteq(yyline + 1, yycolumn + 1); }
4
5  ...

```

- 对非法单词的正则表达式，也和ex2类似

```

1  /* Exception definitions */
2
3  IllegalComment = "(*" ([^*] | "*" + [^\\])*) | ([^\\] | "(" + [^*]*) "*"
4  IllegalOctal  = 0[0-7]*[9|8]+[0-9]*
5  IllegalInteger = {MyInteger} + {Identifier}+

```

4.2.2 语法分析

4.2.2.1 环境表

- env 表示一个环境，用于存储声明及其作用域链

```

1  package parser.env;
2
3  import java.util.*;
4  import exceptions.*;
5  import parser.type.*;

```

```

6
7  /**
8   * 表示一个环境，用于存储声明及其作用域链。
9   */
10 public class Env {
11
12     private Hashtable<String, Declaration> decls;
13     protected Env prev;
14
15     /**
16      * 创建一个基于当前环境的新嵌套环境。
17      *
18      * @param cur 当前环境
19      */
20     public Env(Env cur) {
21         decls = new Hashtable<String, Declaration>();
22         prev = cur;
23     }
24
25     /**
26      * 向环境中添加声明。
27      *
28      * @param s 声明的词素
29      * @param t 声明对象
30      * @throws OberonException 如果已存在同名声明
31      */
32     public void putDecl(String s, Declaration t) throws OberonException {
33         if (decls.containsKey(s.toLowerCase()))
34             throw new OberonException();
35         decls.put(s.toLowerCase(), t);
36     }
37
38     /**
39      * 从环境中获取声明。
40      *
41      * @param s 声明的词素
42      * @return 声明对象，若不存在则返回null
43      */
44     public Declaration getDecl(String s) {
45         String key = s.toLowerCase();
46         for (Env e = this; e != null; e = e.prev) {
47             Declaration found = e.decl.get(key);
48             if (found != null)
49                 return found;
50         }
51         return null;
52     }
53 }

```

- `declaration` 表示一个声明，包括标识符、类型和扩展属性等

```

1  /**
2   * 构造一个声明对象。
3   * @param id 标识符
4   * @param type 类型
5   * @param extAttrs 扩展属性
6   */
7  public Declaration(Identifier id, Type type, Hashtable<String, Object> extAttrs) {
8      this.id = id;
9      this.lexeme = id.getLexeme();
10     this.line = id.getLine();
11     this.column = id.getColumn();
12     this.type = type;
13     this.extAttrs = extAttrs == null ? new Hashtable<String, Object>() : extAttrs;
14 }

```

- `call` 是语法分析过程中记录调用的主要类，表示一次过程调用，包括过程标识符、签名和环境

```

1  /**
2  * 构造一个过程调用对象。
3  * @param id 过程标识符
4  * @param params 参数类型列表
5  * @param env 当前环境
6  */
7  public Call(Identifier id, ArrayList<Type> params, Env env) {
8      this.id = id;
9      this.signature = ProcedureType.rawSig(params);
10     this.env = env;
11 }

```

4.2.2.2 递归下降预测分析

- `AddSeq` 是一个接口：
 - 语法分析过程中parser获取一个方法 `add` 的引用，该方法能将语句添加到它当前正在构建的流程图结构中
 - 通过方法引用创建这个接口对象，语法分析过程中可以将这个引用以 `AddSeq` 对象的形式传递给 `stat_block` 和 `stat_seq` 等解析方法
 - 每个 `stat` 方法都只会调用自己接收到的 `add`

```

1  package parser;
2
3  import flowchart.*;
4
5  public interface AddSeq {
6      public AbstractStatement add(AbstractStatement arg0);
7  }

```

- `NextToken` 也是一个类似的接口，只不过它的用处只有替代词法分析器的 `next_token`

```

1  package parser;
2
3  import exceptions.*;
4  import scanner.token.*;
5
6  public interface NextToken {
7      public Token nextToken() throws java.io.IOException, OberonException;
8  }
9

```

- `parser` 按照实验要求进行递归下降预测分析
 - 文法的每一非终结符号应对应着一个递归子程序，开始符号则对应着其中的主程序
 - 由向前看符号Lookahead决定分支动作
 - 每一个继承属性对应一个形式参数，所有综合属性对应返回值，子结点的每一属性对应一个局部变量
 - 翻译模式中产生式右部的终结符号、非终结符号与语义动作分别执行匹配、递归调用和嵌入代码等动作
- 这里的具体实现和上述 `proc` 规则一致：
 - 初始化 `g_proc` 作为局部流程图，初始化局部符号表
 - 继承属性作为参数传递表示在全局模块流程图 `g_mod` 中，为当前过程创建一个新的过程流程图 `g_proc`

```

1  /**
2  * 解析单个过程声明。
3  */
4  private void proc_decl(flowchart.Module g_mod) throws IOException, OberonException {
5      // 递归下降分析 proc_decl
6
7
8      if (Lookahead.equals(Keyword.symProcedure())) {
9          // 应用产生式 eq1
10

```



```

11     Env cur = env;
12     env = new Env(env);
13
14     _r_proc_head head = proc_head();
15
16     match(Symbol.semicol(), Symbol.class);
17
18     flowchart.Procedure g_proc = g_mod.add(head.id.getLexeme());
19
20     proc_body(g_mod, g_proc, head.id);
21
22     env = cur;
23     try {
24         env.putDecl(head.id.getLexeme(), head.decl);
25     } catch (OberonException e) {
26         Declaration prevDecl = env.getDecl(head.id.getLexeme());
27         throw new IdentifierAlreadyExists(head.id, prevDecl.getId());
28     }
29
30     // 分析结果返回 proc_decl
31 }
32 else {
33     throw new UnexpectedToken(keyword.symProcedure(), lookahead);
34 }
35 }

```

- 调用 `stat_block(g_proc::add)` 表示将新创建的 `g_proc` 的 `add` 方法传递下去，确保该过程体内的语句被正确地添加到自己的流图，其中 `add` 是 java 提供的接口 `interface`

```

1  /**
2   * 解析过程体。
3   */
4  private void proc_body(flowchart.Module g_mod, flowchart.Procedure g_proc, Identifier decl_id)
5  throws IOException, OberonException {
6      // 递归下降分析 proc_body
7
8      if (
9          lookahead.equals(keyword.symConst()) ||
10         lookahead.equals(keyword.symType()) ||
11         lookahead.equals(keyword.symVar()) ||
12         lookahead.equals(keyword.symProcedure()) ||
13         lookahead.equals(keyword.symBegin()) ||
14         lookahead.equals(keyword.symEnd())
15     ) {
16         // 应用产生式 eq1
17
18         decls(g_mod);
19
20         stat_block(g_proc::add);
21
22         match(keyword.symEnd(), keyword.class);
23
24         Identifier id = match(new Identifier(), Identifier.class);
25         if (!decl_id.getLexeme().toLowerCase().equals(id.getLexeme().toLowerCase())) {
26             throw new MismatchedDeclaration("PROCEDURE", decl_id, id);
27         }
28
29         // 分析结果返回 proc_body
30     }
31     else {
32         throw new UnexpectedToken(new Token[] { keyword.symConst(), keyword.symType(),
33             keyword.symVar(), keyword.symProcedure(), keyword.symBegin(), keyword.symEnd() }, lookahead);
34     }
35 }

```


4.2.3 语义检查


- 由于本次实验的主要任务是绘制流图，所以实际上语义动作的实现已经在递归下降预测的语法分析过程中已经完成了、即在分析的过程中划分基本块并给出流图，这与过程、分支等语句是紧密相关的，所以语义分析实际上在上面已经近乎完备了
- 为了利用上面已经给出的各种异常和类型匹配规则，这里的语义分析仅用作类型匹配、参数匹配等语义检查
- 比如 `or` 的操作数必须是 `BooleanType`，若不符合则抛出异常

```
1  else if (lookahead.equals(Operator.or())) {
2      // 应用产生式 eq3
3
4      Operator op = match(Operator.or(), Operator.class);
5
6      _r_term t = term();
7
8      if (!(l_type instanceof BooleanType) || !(t.type instanceof BooleanType)) {
9          throw new UnexpectedType(op, new BooleanType(), new BooleanType(), l_type, t.type);
10     }
11
12     Type post_l_type = new BooleanType();
13     String post_l_text = l_text + " OR " + t.text;
14
15     _r_post_term postTerm = post_term(post_l_type, post_l_text);
16
17     // 分析结果返回 post_term
18     return new _r_post_term(postTerm.type, postTerm.text);
19 }
```


4.2.3 运行测试

4.2.3.0 正确版本





分解质因数的函数流程图太大了，上面只有start、下面只有end被覆盖了



4.2.3.1 变异版本

由于实验要求没有要求，这里就不一一展示了，但是前述的各种异常也能被正确抛出

```

C:\WINDOWS\system32\cmd.exe
Running Testcase 007: MissingRightParenthesisException
=====
exceptions.UnexpectedToken: Unexpected token 'semicolon' at line 59, column 17. Expected 'comma, or rpar'.
请按任意键继续. . .

C:\WINDOWS\system32\cmd.exe
Running Testcase 008: MissingLeftParenthesisException
=====
exceptions.UnexpectedToken: Unexpected token 'id' at line 9, column 21. Expected 'lpar, or semicolon'.
请按任意键继续. . .
  
```

4.3 语法分析讨论

4.3.1 分析技术的简单性、调试便利性

4.3.1.1 递归下降预测分析

4.3.1.1.1 优点

- 递归下降法更加简单，因为其核心思想非常直观，即为每个非终结符编写一个对应的分析函数
- 递归下降法整个分析程序就是这些函数之间的相互调用，其结构与语言的BNF范式高度同构
- 递归下降法这种直接的映射关系使得代码容易理解和维护，也非常容易调试
- 实际实验过程中，递归下降法调试过程非常轻松，因为可以在关键函数处设置断点，查看调用栈，跟踪程序的执行流程，从而快速定位语法匹配错误，大部分IDE也能正确抛出异常定位

4.3.1.1.2 缺点

- 递归下降法简单性来自于在改造文法时的各种规范和限制，即要求文法必须是LL(1)的
- 递归下降法改造过程中最复杂的部分是消除左递归和提取左公因子，这个过程本身容易出错，且修改后的文法可读性会变得较差、语法规则的构造也就变得更加复杂

4.3.1.2 LR分析

优点

- LR分析的分析算法和驱动表等具体实现都是标准化的，通常不需要手动编写，而是使用前几个实验给出的Yacc、Bison、CUP等工具自动生成，实际编写过程中也比较简单
- LR分析能力比LL更强，这也就意味着文法需要的调整更小，而更直观、更简单的文法在语法分析、语义分析的设计中会更加简单

缺点

- LR分析器的内部工作机制非常抽象和复杂，它基于一个下推自动机的状态转换，对于一个错误往往很难直观地看报告错误定位
- LR分析对应的工具包含状态机报告、比如 `.cup` 文件生成的 `parser.out`，但是还是需要理解项集、闭包和状态转换，才能正确的解决文法冲突
- LR分析调试运行时错误也极为困难，因为调试器展示的是一个不断变化的状态栈和输入流，而不是直观的函数调用

4.3.2 分析技术的通用性、能处理的语言范围

4.3.2.1 递归下降预测分析

缺点

- 递归下降通用性较差，标准的递归下降预测分析器只能处理LL(1) 文法，这是一个相对严格的文法，是大部分常见文法的真子集
- 递归下降都无法直接处理左递归，必须进行改造，这包括许多在程序设计语言中非常自然的语法结构，如左递归的表达式 `expr -> expr + term`

4.3.2.2 LR分析

优点

- LR分析通用性极强，能够处理比 LL(1) 大得多的文法类，几乎所有无二义的上下文无关文法都可以用LR(1)来分析
- 实践中常用的 LALR(1)虽然比 LR(1) 稍弱，但仍能覆盖绝大多数程序设计语言的语法结构
- LR分析可以直接处理左递归，这在定义表达式、列表等语法时非常方便

4.3.3 语义动作与语法制导翻译的便利性

4.3.3.1 递归下降预测分析:

优点

- 递归下降预测嵌入语义动作非常灵活，本质上分析器就是程序员手写的代码，所以可以在分析函数的任何位置插入任意的语义动作
- 递归下降可以便利地处理需要在规则中间执行的动作

4.3.3.2 LR分析

优点

- LR分析的语义动作的嵌入位置只能附加在产生式的末尾当分析器完成一次归约操作时，相应的动作才被执行
- LR也可以通过处理嵌入在产生式中间的语义动作，即使用marker机制
- 上述这种方式非常规整，使得语法和语义的耦合更加清晰，便于编写语义动作

缺点

- LR分析的语义动作的嵌入位置只能附加在产生式的末尾当分析器完成一次归约操作时，相应的动作才被执行
- LR分析的语义动作也是高度受限的，在处理继承属性相关等嵌入在产生式内部的语义动作时比较受限
- LR分析使用marker方法会导致文法变得复杂，在语义设计上也会更加抽象

4.3.4 出错恢复的实现难易度

4.3.4.1递归下降预测分析:

缺点

- 递归下降分析实现高质量的出错恢复比较困难，通常是启发式和特设的
- 递归下降分析可以考虑实现恐慌模式，当在某个分析函数中发现不匹配的记号时，就向前扫描输入流，直到找到一个由该函数所能处理的同步记号syn，然后继续分析
- 递归下降分析设计完备的同步记号集需要对语言有深入的理解，相对抽象和复杂

4.3.4.2 LR 分析:

优点

- LR分析虽然同样复杂，但其框架为系统化的出错恢复提供了更好的支持，对每次归约都是对句柄局部进行操作、此时进行错误判断和出错恢复要简单一些
- LR分析有良好的生态支持，像Yacc/Bison这样的工具提供了一个特殊的 `error` 终结符，程序员可以在文法中明确地编写包含 `error` 的产生式，来定义当错误发生时，分析器应该如何跳转状态、丢弃多少记号，并从哪个语法结构开始恢复分析
- LR分析整体上实现更健壮、更可预测，出错恢复更为简单

4.3.5 分析表大小

4.3.5.1 递归下降预测分析:

优点

- 递归下降分析是手动编写的递归程序，LL1分析法是表格驱动的方式实现
- LL(1)分析表的大小为 $|V_n| \times |V_t|$ 非终结符数量 \times 终结符数量
- LL(1)分析表大小相对较小，且表内部一般比较稀疏

4.3.5.2 LR 分析:

缺点

- LR分析表的大小与状态数量成正比，状态数量可能非常大，对于复杂的文法状态数可能难以估量
- LR(1)分析表可能包含数千个状态，占用相当大的空间
- LALR(1) 通过合并状态，显著减小了分析表的大小，但其分析表通常也比同等文法的LL(1)表要大得多

4.3.6 分析速度

- LL(1)和LR分析表格驱动的分析器都非常高效，核心操作都是查表和栈操作，时间复杂度与输入代码的长度成线性关系 $O(n)$
- 递归下降分析器由于手写函数调用的开销，可能比高度优化的表格驱动LR分析器慢，一定程度上取决于程序员的代码水平，但整体上区别应该是微乎其微的